

# Smarandache Sequences: Explorations and Discoveries with a Computer Algebra System\*

Paulo D. F. Gouveia  
 pgouveia@ipb.pt  
 Technology and Manag. School  
 Bragança Polytechnic Institute  
 5301-854 Bragança, Portugal

Delfim F. M. Torres  
 delfim@mat.ua.pt  
 Department of Mathematics  
 University of Aveiro  
 3810-193 Aveiro, Portugal

## Abstract

We study Smarandache sequences of numbers, and related problems, via a Computer Algebra System. Solutions are discovered, and some conjectures presented.

**Mathematics Subject Classification 2000.** 11B83, 11-04, 68W30.

**Keywords.** Smarandache square/cube/prime digital subsequence, Smarandache Lucas/Fibonacci partial digital subsequence, Smarandache odd/even/prime/Fibonacci sequence, Smarandache function, Smarandache 2-2-additive/2-2-subtractive/3-3-additive relationship, Smarandache partial perfect additive sequence, Smarandache bad numbers, Smarandache prime conjecture, mathematical experimentation.

## 1 Introduction

After a good look on the *Mathematics Unlimited—2001 and Beyond* [5], which addresses the question of the future of Mathematics in the new millennium, it is impossible not to get the deep impression that *Computing* will be an integral part of many branches of Mathematics. If it is true that in the XXst century Mathematics has contributed, in a fundamental way, to technology, now, in the XXIst century, the converse seems to be also a possibility. For perspectives on the role of Computing in Mathematics (and the other way around) see [2, 4, 9].

Many powerful and versatile Computer Algebra Systems are available nowadays, putting at our disposal sophisticated environments of mathematical and scientific computing. They comprise both numerical and symbolic computation through high-level and expressive languages, close to the mathematical one. A

---

\*To appear in the 2003 December issue of the *Smarandache Notions Journal*.

large quantity of mathematical knowledge is already available in these scientific systems, providing efficient mathematical methods to perform the desired calculations. This has two important implications: they spare one a protracted process of programming and debugging, so common to the more conventional computer languages; they permit us to write few lines of code, and simpler programs, more declarative in nature. Our claim is that explorations with such tools can develop intuition, insight, and better qualitative understanding of the nature of the problems. This can greatly assist the proof of mathematical results (see an example in Section § 2.1 below).

It is our aim to show that computer-assisted algebra can provide insight and clues to some open questions related to special sequences in Number Theory. Number Theory has the advantage of being easily amenable to computation and experimentation. Explorations with a Computer Algebra System will allow us to produce results and to formulate conjectures. We illustrate our approach with the mathematics Maple system (all the computational processing was carried with Maple version 8, on an AMD Athlon(TM) 1.66 GHz machine), and with some of the problems proposed by the Romanian mathematician Florentin Smarandache.

Maple was originated more than two decades ago, as a project of the Symbolic Computation Group of the University of Waterloo, Ontario. It is now a registered trademark product of Waterloo Maple Inc. We refer the reader to [19, 13] for a gentle introduction to Maple. For a good account on the Smarandache collection of problems, and for a biography of F. Smarandache, see [10].

We invite and exhort readers to convert our mathematical explorations in the language of their favorite Computer Algebra System; to optimize the algorithms (we have followed the didactic approach, without any attempt of code optimization); and to obtain the results for themselves. The source be with you.

## 2 Smarandache Digital Subsequences

We begin by considering sequences of natural numbers satisfying some given property together with all their digits.

### 2.1 Smarandache p-digital subsequences

We are interested in the following *Smarandache p-digital subsequences*. Let  $p \geq 2$ . From the sequence  $\{n^p\}$ ,  $n \in \mathbb{N}_0$ , we select those terms whose digits are all perfect  $p$ -powers. For  $p = 2$  we obtain the *Smarandache square-digital subsequence*: we select only those terms of the sequence  $\{n^2\}_{n=0}^{\infty}$  whose digits belong to the set  $\{0, 1, 4, 9\}$ . With the Maple definitions

```
> pow := (n,p) -> seq(i^p,i=0..n):
> perfectPow := (n,p) -> evalb(n = iroot(n,p)^p):
> digit := (n,num) -> irem(iquo(num,10^(length(num)-n)),10):
> digits := n -> map(digit,[$1..length(n)],n):
> digPerfectPow :=
```

```
> (n,p) -> evalb(select(perfectPow,digits(n),p) = digits(n)):
```

the Smarandache square-digital subsequence is easily obtained:

```
> ssds := n -> select(digPerfectPow,[pow(n,2)],2):
```

We now ask for all the terms of the Smarandache square-digital subsequence which are less or equal than  $10000^2$ :

```
> ssds(10000);
```

```
[0, 1, 4, 9, 49, 100, 144, 400, 441, 900, 1444, 4900, 9409, 10000, 10404, 11449,
 14400, 19044, 40000, 40401, 44100, 44944, 90000, 144400, 419904, 490000,
  491401, 904401, 940900, 994009, 1000000, 1004004, 1014049, 1040400,
 1100401, 1144900, 1440000, 1904400, 1940449, 4000000, 4004001, 4040100,
 4410000, 4494400, 9000000, 9909904, 9941409, 11909401, 14010049, 14040009,
 14440000, 19909444, 40411449, 41990400, 49000000, 49014001, 49140100,
 49999041, 90440100, 94090000, 94109401, 99400900, 99940009, 100000000]
```

In [3, 18] one finds the following question:

“Disregarding the square numbers of the form  $N \times 10^{2k}$ ,  $k \in \mathbb{N}$ ,  $N$  also a perfect square number, how many other numbers belong to the Smarandache square-digital subsequence?”

From the obtained 64 numbers of the Smarandache square-digital subsequence, one can see some interesting patterns from which one easily guess the answer.

**Theorem 1.** *There exist an infinite number of terms on the Smarandache square-digital subsequence which are not of the form  $N \times 10^{2k}$ ,  $k \in \mathbb{N}$ ,  $N$  a perfect square number.*

Theorem 1 is a straightforward consequence of the following Lemma.

**Lemma 2.** *Any number of the form  $(10^{k+1} + 4) \times 10^{k+1} + 4$ ,  $k \in \mathbb{N}_0$  (144, 10404, 1004004, 100040004, ...), belong to the Smarandache square-digital subsequence.*

*Proof.* Lemma 2 follows by direct calculation:

$$(10^{k+1} + 2)^2 = (10^{k+1} + 4) \times 10^{k+1} + 4.$$

□

We remark that from the analysis of the list of the first 64 terms of the Smarandache square-digital subsequence, one easily finds other possibilities to prove Theorem 1, using different but similar assertions than the one in Lemma 2. For example, any number of the form  $(10^{k+2} + 14) \times 10^{k+2} + 49$ ,  $k \in \mathbb{N}_0$  (11449, 1014049, 100140049, ...), belong to the Smarandache square-digital subsequence:

$$(10^{k+2} + 7)^2 = (10^{k+2} + 14) \times 10^{k+2} + 49.$$

Other possibility, first discovered in [12], is to use the pattern  $(4 \times 10^{k+1} + 4) \times 10^{k+1} + 1$ ,  $k \in \mathbb{N}_0$  (441, 40401, 4004001, ...), which is the square of  $2 \times 10^{k+1} + 1$ .

Choosing  $p = 3$  we obtain the *Smarandache cube-digital subsequence*.

```
> scds := n -> select(digPerfectPow, [pow(n, 3)], 3):
```

Looking for all terms of the Smarandache cube-digital subsequence which are less or equal than  $10000^3$  we only find the trivial ones:

```
> scds(10000);
```

```
[0, 1, 8, 1000, 8000, 1000000, 8000000, 1000000000, 8000000000, 1000000000000]
```

We offer the following conjecture:

**Conjecture 3.** *All terms of the Smarandache cube-digital subsequence are of the form  $D \times 10^{3k}$  where  $D \in \{0, 1, 8\}$  and  $k \in \mathbb{N}_0$ .*

Many more Smarandache digital subsequences have been introduced in the literature. One good example is the Smarandache prime digital subsequence, defined as the sequence of prime numbers whose digits are all primes (see [18]).

Terms of the Smarandache prime digital subsequence are easily discovered with the help of the Maple system. Defining

```
> primeDig := n -> evalb(select(isprime, digits(n)) = digits(n)):
> spds := n -> select(primeDig, [seq(ithprime(i), i=1..n)]):
```

we find that 189 of the first 10000 prime numbers belong to the Smarandache prime digital subsequence:

```
> nops(spds(10000));
```

189

## 2.2 Smarandache p-partial digital subsequences

The *Smarandache p-partial digital subsequence* is defined by scrolling through a given sequence  $\{a_n\}$ ,  $n \geq 0$ , defined by some property  $p$ , and selecting only those terms which can be partitioned in groups of digits satisfying the same property  $p$  (see [3]). For example, let us consider  $\{a_n\}$  defined by the recurrence relation  $a_n = a_{n-1} + a_{n-2}$ . One gets the *Lucas sequence* by choosing the initial conditions  $a_0 = 2$  and  $a_1 = 1$ ; the *Fibonacci sequence* by choosing  $a_0 = 0$  and  $a_1 = 1$ . The Smarandache *Lucas*-partial digital subsequence and the Smarandache *Fibonacci*-partial digital subsequence are then obtained selecting from the respective sequences only those terms  $n$  for which there exist a partition of the digits in three groups ( $n = g_1g_2g_3$ ) with the sum of the first two groups equal to the third one ( $g_1 + g_2 = g_3$ ).

In [3, 17, 16] the following questions are formulated:

“Is 123 ( $1+2 = 3$ ) the only Lucas number that verifies a Smarandache type partition?”

“We were not able to find any Fibonacci number verifying a Smarandache type partition, but we could not investigate large numbers; can you?”

Using the following procedure, we can verify if a certain number  $n$  fulfills the necessary condition to belong to the Smarandache *Lucas/Fibonacci*-partial digital subsequence, i.e., if  $n$  can be divided in three digit groups,  $g_1g_2g_3$ , with  $g_1+g_2=g_3$ .

```
> spds:=proc(n)
>   local nd1, nd2, nd3, nd, g1, g2, g3:
>   nd:=length(n);
>   for nd3 to nd-2 do
>     g3:=irem(n,10^nd3);
>     if length(g3)*2>nd then break; fi;
>     for nd1 from min(nd3,nd-nd3-1) by -1 to 1 do
>       nd2:=nd-nd3-nd1;
>       g1:=iquo(n,10^(nd2+nd3));
>       g2:=irem(iquo(n,10^nd3), 10^nd2);
>       if g2>=g3 then break;fi;
>       if g1+g2=g3 then printf("%d (%d+%d=%d)\n",n,g1,g2,g3);fi;
>     od;
>   od:
> end proc:
```

Now, we can compute the first  $n$  terms of the Lucas sequence, using the procedure below.

```
> lucas:=proc(n)
>   local L, i:
>   L:=[2, 1]:
>   for i from 1 to n-2 do L:=[L[],L[i]+L[i+1]]:od:
> end proc:
```

With  $n = 20$  we get the first twenty Lucas numbers

```
> lucas(20);
[2, 1, 3, 4, 7, 11, 18, 29, 47, 76, 123, 199, 322, 521, 843, 1364, 2207, 3571, 5778, 9349]
```

Let  $L$  be the list of the first 6000 terms of the Lucas sequence:

```
> L:=lucas(6000):
(elapsed time: 1.9 seconds) 1
```

It is interesting to remark that the 6000<sup>th</sup> element has 1254 digits:

---

<sup>1</sup>The most significant time calculations are showed, in order to give an idea about the involved computation effort.

```
> length(L[6000]);
```

1254

The following Maple command permit us to check which of the first 3000 elements belong to a Smarandache *Lucas*-partial digital subsequence.

```
> map(spds, L[1..3000]):
```

```
123 (1+2=3)
20633239 (206+33=239)
```

```
(elapsed time: 7h50m)
```

As reported in [15], only two of the first 3000 elements of the Lucas sequence verify a Smarandache type partition: the 11<sup>th</sup> and 36<sup>th</sup> elements.

```
> L[11], L[36];
```

123, 20633239

We now address the following question: Which of the next 3000 elements of the Lucas sequence belong to a Smarandache *Lucas*-partial digital subsequence?

```
> map(spds, L[3001..6000]):
```

```
(elapsed time: 67h59m)
```

The answer turns out to be *none*: no number, verifying a Smarandache type partition, was found between the 3001<sup>th</sup> and the 6000<sup>th</sup> term of the Lucas sequence.

The same kind of analysis is easily done for the Fibonacci sequence. We compute the terms of the Fibonacci sequence using the pre-defined function `fibonacci`:

```
> with(combinat, fibonacci):
> [seq(fibonacci(i), i=1..20)];
```

```
[1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144, 233, 377, 610, 987, 1597, 2584, 4181, 6765]
```

Although the 6000<sup>th</sup> Fibonacci number is different from the 6000<sup>th</sup> Lucas number

```
> evalb(fibonacci(6000) = L[6000]);
```

*false*

they have the same number of digits

```
> length(fibonacci(6000));
```

In order to identify which of the first 3000 Fibonacci numbers belong to the Smarandache *Fibonacci*-partial digital subsequence, we execute the following short piece of Maple code:

```
> map(spds, [seq(fibonacci(i), i=1..3000)]):
832040 (8+32=040)
(elapsed time: 8h32m)
```

This is in consonance with the result reported in [15]: only one number, among the first 3000 numbers of the Fibonacci sequence, verifies a Smarandache type partition – the 30<sup>th</sup> one.

```
> fibonacci(30);
```

832040

As before, with respect to the Lucas sequence, we now want to know which of the next 3000 numbers of the Fibonacci sequence belong to the Smarandache *Fibonacci*-partial digital subsequence.

```
> map(spds, [seq(fibonacci(i), i=3001..6000)]):
(elapsed time: 39h57m)
```

Similarly to the Lucas case, no number, verifying a Smarandache type partition, was found between the 3001<sup>th</sup> and the 6000<sup>th</sup> term of the Fibonacci sequence.

### 3 Smarandache Concatenation-Type Sequences

Let  $\{a_n\}$ ,  $n \in \mathbb{N}$ , be a given sequence of numbers. The Smarandache concatenation sequence associated to  $\{a_n\}$  is a new sequence  $\{s_n\}$  where  $s_n$  is given by the concatenation of all the terms  $a_1, \dots, a_n$ . The concatenation operation between two numbers  $a$  and  $b$  is defined as follows:

```
> conc := (a,b) -> a*10^length(b)+b:
```

In this section we consider four different Smarandache concatenation-type subsequences: the odd, the even, the prime, and the Fibonacci one.

```
> oddSeq := n -> select(type, [seq(i, i=1..n)], odd):
> evenSeq := n -> select(type, [seq(i, i=1..n)], even):
> primeSeq := n -> [seq(ithprime(i), i=1..n)]:
> with(combinat, fibonacci):
> fibSeq := n -> [seq(fibonacci(i), i=1..n)]:
```

```

> # ss = Smarandache Sequence
> ss := proc(F,n)
>   local L, R, i:
>   L := F(n):
>   R := array(1..nops(L)): R[1] := L[1]:
>   for i from 2 while i <= nops(L) do
>     R[i]:=conc(R[i-1],L[i]):
>   end do:
>   evalm(R):
> end proc:

```

Just to illustrate the above definitions, we compute the first five terms of the Smarandache odd, even, prime, and Fibonacci sequences:

```

> ss(oddSeq,10);

[1, 13, 135, 1357, 13579]

> ss(evenSeq,10);

[2, 24, 246, 2468, 246810]

> ss(primeSeq,5);

[2, 23, 235, 2357, 235711]

> ss(fibSeq,5);

[1, 11, 112, 1123, 11235]

```

Many interesting questions appear when one try to find numbers among the terms of a Smarandache concatenation-type sequence with some given property. For example, it remains an open question to understand how many primes are there in the odd, prime, or Fibonacci sequences. Are they infinitely or finitely in number? The following procedure permit us to find prime numbers in a certain Smarandache sequence.

```

> ssPrimes := proc(F,n)
>   local ar, i:
>   ar := select(isprime,ss(F,n)):
>   convert(ar,list):
> end proc:

```

There are five prime numbers in the first fifty terms of the Smarandache odd sequence;

```

> nops(ssPrimes(oddSeq,100));

```

5

five prime numbers in the first two hundred terms of the Smarandache prime sequence;

```
> nops(ssPrimes(primeSeq,200));
```

5

and two primes (11 and 1123) in the first one hundred and twenty terms of the Smarandache Fibonacci sequence.

```
> ssPrimes(fibSeq,120);
```

[11,1123]

It is clear that only the first term of the Smarandache even sequence is prime. One interesting question, formulated in [1, Ch. 2], is the following:

“How many elements of the Smarandache even sequence are twice a prime?”

A simple search with Maple shows that 2468101214 is the only number twice a prime in the first four hundred terms of the Smarandache even sequence (the term 400 of the Smarandache even sequence is a number with 1147 decimal digits).

```
> ssTwicePrime := proc(n)
>   local ar, i:
>   ar := select(i->isprime(i/2),ss(evenSeq,n)):
>   convert(ar,list):
> end proc:
> ssTwicePrime(800);
```

[2468101214]

## 4 Smarandache Relationships

We now consider the so called *Smarandache function*. This function  $S(n)$  is important for many reasons (cf. [10, pp. 91–92]). For example, it gives a necessary and sufficient condition for a number to be prime:  $p > 4$  is prime if, and only if,  $S(p) = p$ . *Smarandache numbers* are the values of the Smarandache function.

### 4.1 Sequences of Smarandache numbers

The *Smarandache function* is defined in [16] as follows:  $S(n)$  is the smallest positive integer number such that  $S(n)!$  is divisible by  $n$ . This function can be defined in Maple by the following procedure:

```
> S:=proc(n)
>   local i, fact:
>   fact:=1:
>   for i from 2 while irem(fact, n)<>0 do
```

```

>   fact:=fact*i:
>   od:
>   return i-1:
> end proc:

```

The first terms of the Smarandache sequence are easily obtained:

```

> seq(S(n),n=1..20);

```

1, 2, 3, 4, 5, 3, 7, 4, 6, 5, 11, 4, 13, 7, 5, 6, 17, 6, 19, 5

A sequence of  $2k$  Smarandache numbers satisfy a Smarandache  $k$ - $k$  additive relationship if

$$S(n) + S(n+1) + \dots + S(n+k-1) = S(n+k) + S(n+k+1) + \dots + S(n+2k-1).$$

In a similar way, a sequence of  $2k$  Smarandache numbers satisfy a Smarandache  $k$ - $k$  subtractive relationship if

$$S(n) - S(n+1) - \dots - S(n+k-1) = S(n+k) - S(n+k+1) - \dots - S(n+2k-1).$$

In [3, 17] one finds the following questions:

“How many quadruplets verify a Smarandache 2-2 additive relationship?”

“How many quadruplets verify a Smarandache 2-2 subtractive relationship?”

“How many sextuplets verify a Smarandache 3-3 additive relationship?”

To address these questions, we represent each of the relationships by a Maple function:

```

> add2_2:=(V,n)->V[n]+V[n+1]=V[n+2]+V[n+3]:
> sub2_2:=(V,n)->V[n]-V[n+1]=V[n+2]-V[n+3]:
> add3_3:=(V,n)->V[n]+V[n+1]+V[n+2]=V[n+3]+V[n+4]+V[n+5]:

```

We compute the first 10005 numbers of the Smarandache sequence:

```

> SSN:=[seq(S(i),i=1..10005)]:

```

```

(elapsed time: 59m29s)

```

With the following procedure, we can identify all the positions in the sequence  $V$  that verify the relationship  $F$ .

```

> verifyRelation:=proc(F,V)
>   local i, VR: VR:=[]:
>   for i to nops(V)-5 do
>     if F(V,i) then VR:=[VR[], i]: fi:
>   od:
>   return VR;
> end proc:

```

We can answer the above mentioned questions for the first 10000 numbers of the Smarandache sequence.

The positions verifying the Smarandache 2-2 additive relationship are:

```

> V1:=verifyRelation(add2_2,SSN);

V1 := [6, 7, 28, 114, 1720, 3538, 4313, 8474]

```

Similarly, we determine the positions verifying the Smarandache 2-2 subtractive relationship,

```

> V2:=verifyRelation(sub2_2,SSN);

V2 := [1, 2, 40, 49, 107, 2315, 3913, 4157, 4170]

```

and the positions verifying the Smarandache 3-3 additive relationship:

```

> V3:=verifyRelation(add3_3,SSN);

V3 := [5, 5182, 9855]

```

The quadruplets associated with the positions V1 (2-2 additive relationship) are given by

```

> map(i->printf("S(%d)+S(%d)=S(%d)+S(%d) [%d+%d=%d+%d]\n",
i,i+1,i+2,i+3,S(i),S(i+1),S(i+2),S(i+3)), V1):

```

```

S(6)+S(7)=S(8)+S(9) [3+7=4+6]
S(7)+S(8)=S(9)+S(10) [7+4=6+5]
S(28)+S(29)=S(30)+S(31) [7+29=5+31]
S(114)+S(115)=S(116)+S(117) [19+23=29+13]
S(1720)+S(1721)=S(1722)+S(1723) [43+1721=41+1723]
S(3538)+S(3539)=S(3540)+S(3541) [61+3539=59+3541]
S(4313)+S(4314)=S(4315)+S(4316) [227+719=863+83]
S(8474)+S(8475)=S(8476)+S(8477) [223+113=163+173]

```

We remark that in M. Bencze's paper [3] only the first three quadruplets were found. The quadruplets associated with the positions V2 (2-2 subtractive relationship) are:

```

> map(i->printf("S(%d)-S(%d)=S(%d)-S(%d) [%d-%d=%d-%d]\n",
i,i+1,i+2,i+3,S(i),S(i+1),S(i+2),S(i+3)), V2):

```

```

S(1)-S(2)=S(3)-S(4) [1-2=3-4]
S(2)-S(3)=S(4)-S(5) [2-3=4-5]
S(40)-S(41)=S(42)-S(43) [5-41=7-43]
S(49)-S(50)=S(51)-S(52) [14-10=17-13]
S(107)-S(108)=S(109)-S(110) [107-9=109-11]
S(2315)-S(2316)=S(2317)-S(2318) [463-193=331-61]
S(3913)-S(3914)=S(3915)-S(3916) [43-103=29-89]
S(4157)-S(4158)=S(4159)-S(4160) [4157-11=4159-13]
S(4170)-S(4171)=S(4172)-S(4173) [139-97=149-107]

```

Only the first two and fourth quadruplets were found in [3]. The following three sextuplets verify a Smarandache 3-3 additive relationship:

```

> map(i->printf("S(%d)+S(%d)+S(%d)=S(%d)+S(%d)+S(%d)
  [%d+%d+%d=%d+%d+%d]\n", i, i+1, i+2, i+3, i+4, i+5,
  S(i), S(i+1), S(i+2), S(i+3), S(i+4), S(i+5)), V3):
S(5)+S(6)+S(7)=S(8)+S(9)+S(10) [5+3+7=4+6+5]
S(5182)+S(5183)+S(5184)=S(5185)+S(5186)+S(5187) [2591+73+9=61+2593+19]
S(9855)+S(9856)+S(9857)=S(9858)+S(9859)+S(9860) [73+11+9857=53+9859+29]

```

Only the first sextuplet was found by M. Bencze's in [3]. For a deeper analysis of these type of relationships, see [6, 8].

## 4.2 An example of a Smarandache partial perfect additive sequence

Let  $\{a_n\}$ ,  $n \geq 1$ , be a sequence constructed in the following way:

$$\begin{aligned}
 a_1 &= a_2 = 1; \\
 a_{2p+1} &= a_{p+1} - 1; \\
 a_{2p+2} &= a_{p+1} + 1.
 \end{aligned}$$

The following Maple procedure defines  $a_n$ .

```

> a:=proc(n)
>   option remember:
>   if (n=1) or (n=2) then return 1:
>   elif type(n, odd) then return a((n-1)/2+1)-1:
>   else return a((n-2)/2+1)+1:
>   fi:
> end proc:

```

In [3] the first 26 terms of the sequence are presented as being

```

> A:=1,1,0,2,-1,1,1,3,-2,0,0,2,1,1,3,5,-4,-2,-1,1,-1,1,1,3,0,2:

```

One easily concludes, as mentioned in [7], that starting from the thirteenth term the above values are erroneous. The correct values are obtained with the help of our procedure:

```
> seq(a(i), i=1..26);
```

1, 1, 0, 2, -1, 1, 1, 3, -2, 0, 0, 2, 0, 2, 2, 4, -3, -1, -1, 1, -1, 1, 1, 3, -1, 1

We prove, for  $1 \leq p \leq 5000$ , that  $\{a_n\}$  is a Smarandache partial perfect additive sequence, that is, it satisfies the relation

$$a_1 + a_2 + \cdots + a_p = a_{p+1} + a_{p+2} + \cdots + a_{2p}. \quad (1)$$

This is accomplished by the following Maple code:

```
> sppasproperty:=proc(n)
>   local SPPAS, p;
>   SPPAS:=seq(a(i), i=1..n);
>   for p from 1 to iquo(n,2) do
>     if evalb(add(SPPAS[i], i=1..p)<>add(SPPAS[i], i=p+1..2*p))
>       then return false;
>     fi;
>   od;
>   return true;
> end proc;
```

```
> sppasproperty(10000);
```

*true*

(elapsed time: 11.4 seconds)

We remark that the erroneous sequence  $A$  does not verify property (1). For example, with  $p = 8$  one gets:

```
> add(A[i], i=1..8)<>add(A[i], i=9..16);
```

$8 \neq 10$

## 5 Other Smarandache Definitions and Conjectures

The Smarandache prime conjecture share resemblances (a kind of dual assertion) with the famous Goldbach's conjecture: "Every even integer greater than four can be expressed as a sum of two primes".

### 5.1 Smarandache Prime Conjecture

In [3, 17, 16] the so called *Smarandache Prime Conjecture* is formulated: "Any odd number can be expressed as the sum of two primes minus a third prime (not including the trivial solution  $p = p + q - q$  when the odd number is the prime itself)".

We formulate a strong variant of this conjecture, requiring the odd number and the third prime to be different (not including the situation  $p = k + q - p$ ), that is, we exclude the situation addressed by Goldbach's conjecture (where the even integer  $2p$  is expressed as the sum of two primes  $k$  and  $q$ ).

The number of times each odd number can be expressed as the sum of two primes minus a third prime, are called *Smarandache prime conjecture numbers*. It seems that none of them are known (cf. [3]). Here we introduce the notion of *strong Smarandache n-prime conjecture numbers*: the number of possibilities that each positive odd number can be expressed as a sum of two primes minus a third prime, excluding the trivial solution and imposing our requirement that the odd number and the third prime must be different, using all possible combinations of the first  $n$  primes.

Given  $n$ , the next procedure determines such numbers for all positive odd integers less or equal than  $lim$ .

```

> spcn:=proc(lim, n)
>   local y, z, i, primos, num, mat:
>   mat:=array(1..lim, 1..2, [seq(['?', 0], i=1..lim)]):
>   primos:=seq(ithprime(i), i=1..n);
>   for i from 1 to n do
>     for y in [primos[i..n]] do
>       for z in [primos] do
>         num:=primos[i]+y-z;
>         if (num>=1 and num<=lim and type(num, odd) and
>           z<>primos[i] and z<>y and z<>num) then
>           if mat[num, 2]=0 then mat[num, 1]:=[primos[i], y, z]:
>           fi:
>           mat[num, 2]:=mat[num, 2]+1;
>         fi:
>       od:
>     od:
>   od:
>   for i by 2 to lim do
>     if mat[i, 2]=0 then printf("%d=? (0 possibilities)\n", i):
>     else printf("%d=%d+%d-%d (%d possibilities)\n", i,
>               op(mat[i, 1]), mat[i, 2]):
>     fi:
>   od:
>   evalm(mat):
> end proc:

```

All positive odd numbers less or equal than 19 can be expressed according to the conjecture, using only the first six primes.<sup>2</sup>

```

> spcn(19,6):

```

---

<sup>2</sup>For each number, only one of the possibilities is showed.

```

1=2+2-3 (6 possibilities)
3=5+5-7 (3 possibilities)
5=3+13-11 (2 possibilities)
7=5+5-3 (2 possibilities)
9=3+11-5 (7 possibilities)
11=3+13-5 (3 possibilities)
13=5+11-3 (2 possibilities)
15=5+13-3 (5 possibilities)
17=7+13-3 (3 possibilities)
19=11+11-3 (3 possibilities)

```

(elapsed time: 0.0 seconds)

As expected, if one uses the first 100 primes, the number of distinct possibilities, for which each number can be expressed as in our conjecture, increases.

```
> spcn(19,100):
```

```

1=2+2-3 (1087 possibilities)
3=5+5-7 (737 possibilities)
5=3+13-11 (1015 possibilities)
7=3+17-13 (1041 possibilities)
9=3+11-5 (793 possibilities)
11=3+13-5 (1083 possibilities)
13=3+17-7 (1057 possibilities)
15=3+17-5 (770 possibilities)
17=3+19-5 (1116 possibilities)
19=3+23-7 (1078 possibilities)
(elapsed time: 1.8 seconds)

```

How many odd numbers less or equal to 10000 verify the conjecture?<sup>3</sup>

```
> SPCN1:=spcn(10000,600):
```

(elapsed time: 30m59s)

```
> n:=0: for i by 2 to 10000 do if SPCN1[i,2]>0 then n:=n+1; fi; od: n;
```

4406

Using the first 600 primes, only 4406 of the 5000 odd numbers verify the conjecture. And if one uses the first 700 primes?

```
> SPCN2:=spcn(10000,700):
```

(elapsed time: 49m34s)

---

<sup>3</sup>In the follow *spcn* procedure calls, we removed from its definition the last *for* loop (*spcn* without screen output).

```

> n:=0:
> for i by 2 to 10000 do if SPCN2[i,2]>0 then n:=n+1; fi; od;
> n;

```

5000

Using the first 700 primes, all the odd numbers up to 10000 verify the conjecture. We refer the readers interested in the Smarandache prime conjecture to [14].

## 5.2 Smarandache Bad Numbers

“There are infinitely many numbers that cannot be expressed as the difference between a cube and a square (in absolute value). They are called *Smarandache Bad Numbers(!)*” – see [3].

The next procedure determines if a number  $n$  can be expressed in the form  $n = |x^3 - y^2|$  (i.e., if it is a non Smarandache bad number), for any integer  $x$  less or equal than  $x_{max}$ . The algorithm is based in the following equivalence

$$n = |x^3 - y^2| \Leftrightarrow y = \sqrt{x^3 - n} \quad \vee \quad y = \sqrt{x^3 + n}.$$

For each  $x$  between 1 and  $x_{max}$ , we try to find an *integer*  $y$  satisfying  $y = \sqrt{x^3 - n}$  or  $y = \sqrt{x^3 + n}$ , to conclude that  $n$  is a non Smarandache bad number.

```

> nsbn:=proc(n,xmax)
>   local x, x3:
>   for x to xmax do
>     x3:=x^3;
>     if issqr(x3-n) and x3<>n then return n[x, sqrt(x3-n)];
>     elif issqr(x3+n) then return n[x, sqrt(x3+n)]; fi;
>   od:
>   return n['?', '?'];
> end proc:

```

F. Smarandache [16] conjectured that the numbers 5, 6, 7, 10, 13, 14, ... are probably bad numbers. We now ask for all the non Smarandache bad numbers which are less or equal than 30, using only the  $x$  values between 1 and 19. We use the notation  $n_{x,y}$  to mean that  $n = |x^3 - y^2|$ . For example,  $1_{2,3}$  means that  $1 = |2^3 - 3^2| = |8 - 9|$ .

```

> NSBN:=map(nsbn, [$1..30], 19);

```

$$NSBN := [1_{2,3}, 2_{3,5}, 3_{1,2}, 4_{2,2}, 5_{7,7}, 6_{7,7}, 7_{2,1}, 8_{1,3}, 9_{3,6}, 10_{7,7}, \\ 11_{3,4}, 12_{13,47}, 13_{17,70}, 14_{7,7}, 15_{1,4}, 16_{7,7}, 17_{2,5}, 18_{3,3}, 19_{5,12}, 20_{6,14}, \\ 21_{7,7}, 22_{3,7}, 23_{3,2}, 24_{1,5}, 25_{5,10}, 26_{3,1}, 27_{7,7}, 28_{2,6}, 29_{7,7}, 30_{19,83}]$$

As proved by Maohua Le in [11], we have just shown that 7 and 13 are non Smarandache bad numbers:  $7 = |2^3 - 1^2|$  and  $13 = |17^3 - 70^2|$ . The possible Smarandache bad numbers are:

```
> select(n->evalb(op(1,n)='?'), NSBN);
```

```
[5?,?, 6?,?, 10?,?, 14?,?, 16?,?, 21?,?, 27?,?, 29?,?]
```

Finally, we will determine if any of these eight numbers is a non Smarandache bad number, if one uses all the  $x$  values up to  $10^8$ .

```
> map(nsbm, [5, 6, 10, 14, 16, 21, 27, 29], 10^8);
```

```
[5?,?, 6?,?, 10?,?, 14?,?, 16?,?, 21?,?, 27?,?, 29?,?]
```

```
(elapsed time: 14h30m)
```

From the obtained result, we conjecture that 5, 6, 10, 14, 16, 21, 27, and 29, are bad numbers. We look forward to readers explorations and discoveries.

## Acknowledgments

This work was partially supported by the program PRODEP III/5.3/2003, and by the R&D unit CEOC of the University of Aveiro.

## References

- [1] C. Ashbacher. *Pluckings from the tree of Smarandache sequences and functions*. American Research Press, Lupton, AZ, 1998.
- [2] D. H. Bailey and J. M. Borwein. Experimental mathematics: recent developments and future outlook. In *Mathematics unlimited—2001 and beyond*, pages 51–66. Springer, Berlin, 2001.
- [3] M. Bencze. Smarandache relationships and subsequences. *Smarandache Notions J.*, 11(1-3):79–85, 2000.
- [4] J. M. Borwein and R. M. Corless. Emerging tools for experimental mathematics. *Amer. Math. Monthly*, 106(10):889–909, 1999.
- [5] B. Engquist and W. Schmid, editors. *Mathematics unlimited—2001 and beyond*. Springer-Verlag, Berlin, 2001.
- [6] H. Ibstedt. A brief account on smarandache 2-2 subtractive relationships. *Smarandache Notions J.*, 12(1-3):99–102, 2001.
- [7] H. Ibstedt. On a Smarandache partial perfect additive sequence. *Smarandache Notions J.*, 12(1-3):103–107, 2001.

- [8] H. Ibstedt. Smarandache  $k$ - $k$  additive relationships. *Smarandache Notions J.*, 12(1-3):62–81, 2001.
- [9] P. D. Lax. Mathematics and computing. In *Mathematics: frontiers and perspectives*, pages 417–432. Amer. Math. Soc., Providence, RI, 2000.
- [10] C. T. Le. The most paradoxist mathematician of the world “Florentin Smarandache”. *Bull. Pure Appl. Sci. Sect. E Math. Stat.*, 15(1):81–96, 1996.
- [11] M. Le. A note on the Smarandache bad numbers. *Smarandache Notions J.*, 12(1-3):215–216, 2001.
- [12] M. Le. The reduced Smarandache square-digital subsequence is infinite. *Smarandache Notions J.*, 12(1-3):313–314, 2001.
- [13] K. H. Rosen, J. S. Devitt, T. Vasiga, J. McCarron, E. Murray, and E. Roskos. *Exploring discrete mathematics with Maple*. McGraw-Hill Inc., Boston (MA), 1997.
- [14] F. Russo. On some Smarandache conjectures and unsolved problems. *Smarandache Notions J.*, 12(1-3):172–192, 2001.
- [15] F. Russo. On two problems concerning two Smarandache P-partial digital subsequences. *Smarandache Notions J.*, 12(1-3):198–200, 2001.
- [16] F. Smarandache. Properties of Numbers. *University of Craiova Archives*, 1975. (see also Arizona State University Special Collections, Tempe, AZ, U.S.A.).
- [17] F. Smarandache. Numerology. Oct. 13 2000. Comment: 16 pages. Tables of numbers. Presented to the Pedagogical High School Student Conference in Craiova, 1969. [arXiv.org:math/0010132](https://arxiv.org/abs/math/0010132)
- [18] S. Smith. A set of conjectures on Smarandache sequences. *Smarandache Notions J.*, 11(1-3):86–92, 2000.
- [19] F. Vivaldi. *Experimental mathematics with Maple*. Chapman & Hall/CRC Mathematics. Chapman & Hall/CRC, Boca Raton, FL, 2001.